Lineární algebra

Robert Mařík

27. listopadu 2010

1 Operace s maticemi

Definice matice

Sage code
```
A=matrix([[1,2],[3,-2]])
A
```

$$\begin{pmatrix} 1 & 2 \\ 3 & -2 \end{pmatrix}$$

Sage code
```
B=matrix([[1],[-4]])
B
```

$$\begin{pmatrix} 1 \\ -4 \end{pmatrix}$$

Transponování matice

Sage code
```
[A.transpose(), B.transpose()]
```

$$\left[\begin{pmatrix} 1 & 3 \\ 2 & -2 \end{pmatrix}, \begin{pmatrix} 1 & -4 \end{pmatrix}\right]$$

Násobení matice číslem a součet matic.

Sage code
```
C=matrix([[2,-3],[1,0]])
2*A-C
```

$$\begin{pmatrix} 0 & 7 \\ 5 & -4 \end{pmatrix}$$

Maticový součin.

Sage code
```
A*B
```

$$\begin{pmatrix} -7 \\ 11 \end{pmatrix}$$

Determinant, hodnost, schodovitý tvar a inverzní matice.

Sage code
```
A.determinant()
```

-8

[0] Podporováno grantem FRVŠ 131/2010.

Sage code
```
A.rank()
```

2

Sage code
```
A.echelon_form()
```

$$\begin{pmatrix} 1 & 2 \\ 0 & 8 \end{pmatrix}$$

Sage code
```
A^(-1)
```

$$\begin{pmatrix} \frac{1}{4} & \frac{1}{4} \\ \frac{3}{8} & -\frac{1}{8} \end{pmatrix}$$

2 Soustavy lineárních rovnic

Pro řešení soustav lineárních i nelineárních rovnic je k dispozici příkaz `solve`.

Sage code
```
x,y,z=var('x y z')
solve([x+y == 1, x+y+z == 6, x-3*z == 6],[x,y,z])
```

$[[x = 21, y = (-20), z = 5]]$

Nekonečně mnoho řešení, záslých na jednom parametru (r_1)

Sage code
```
solve([x+y == 1, x+y+z == 6],[x,y,z])
```

$[[x = -r_3 + 1, y = r_3, z = 5]]$

Soustava která nemá žádné řešení.

Sage code
```
solve([x+y+z == 1, x+y+z == 6],[x,y,z])
```

[]

3 Soustavy lineárních rovnic maticově

Nadefinujeme koeficienty levých stran (matice soustavy), vektor pravých stran a vektor neznámých.

Sage code
```
x1,x2,x3=var('x1 x2 x3')
A=matrix([[1,2,0],[2,1,1],[2,1,-1]])
A
```

$$\begin{pmatrix} 1 & 2 & 0 \\ 2 & 1 & 1 \\ 2 & 1 & -1 \end{pmatrix}$$

Sage code
```
b=matrix([[2],[1],[4]])
b
```

$$\begin{pmatrix} 2 \\ 1 \\ 4 \end{pmatrix}$$

Sage code
```
X = matrix([[x1, x2, x3]]).transpose()
X
```

$$\left(\begin{array}{r} x_{1} \\ x_{2} \\ x_{3} \end{array}\right)$$

Násobením matice soustavy a vektoru neznámých obdržíme levé strany soustavy.

Sage code
```
A*X
```

$$\left(\begin{array}{r} x_{1} + 2 \, x_{2} \\ 2 \, x_{1} + x_{2} + x_{3} \\ 2 \, x_{1} + x_{2} - x_{3} \end{array}\right)$$

Sage code
```
html("$ %s = %s $"%(latex(A*X),latex(b)) )
```

$$\left(\begin{array}{r} x_{1} + 2 \, x_{2} \\ 2 \, x_{1} + x_{2} + x_{3} \\ 2 \, x_{1} + x_{2} - x_{3} \end{array}\right) = \left(\begin{array}{r} 2 \\ 1 \\ 4 \end{array}\right)$$

Řešení soustavy rovnic pomocí inverzní matice

Sage code
```
A^(-1)*b
```

$$\left(\begin{array}{r} 1 \\ \frac{1}{2} \\ -\frac{3}{2} \end{array}\right)$$

Sage code
```
A^(-1)*b.n()
```

$$\left(\begin{array}{r} 1.00000000000000 \\ 0.500000000000000 \\ -1.50000000000000 \end{array}\right)$$

Řešení soustavy rovnic pomocí příkazu `solve`. Nejprve musíme seskládat koeficienty s neznámými a s pravými stranami.

Sage code
```
eq=[ (A*X)[i][0] == b[i][0] for i in (0,1,2) ]
eq
```

$[x_{1} + 2 \, x_{2} = 2, 2 \, x_{1} + x_{2} + x_{3} = 1, 2 \, x_{1} + x_{2} - x_{3} = 4]$

Sage code
```
sol=solve(eq,[x1,x2,x3])
sol
```

$$\left[\left[x_{1} = 1, x_{2} = \left(\frac{1}{2}\right), x_{3} = \left(-\frac{3}{2}\right)\right]\right]$$

Numerická aproximace (`rhs` značí pravou stranu a `n` slouží pro numerickou aproximaci)

Sage code
```
[ i.rhs().n() for i in sol[0] ]
```

$[1.00000000000000, 0.500000000000000, -1.50000000000000]$

Rozšířená matice soustavy a její schodovitý tvar.

Sage code

```
D=(A.columns())
D.extend(b.columns())
AA=matrix(D).transpose()
AA
```

$$\left(\begin{array}{rrrr} 1 & 2 & 0 & 2 \\ 2 & 1 & 1 & 1 \\ 2 & 1 & -1 & 4 \end{array}\right)$$

Sage code

```
AA.echelon_form()
```

$$\left(\begin{array}{rrrr} 1 & 2 & 0 & 2 \\ 0 & 3 & 1 & 0 \\ 0 & 0 & 2 & -3 \end{array}\right)$$